
Potrivirea �irurilor  
(c�utarea apari�iei unui sub�ir(model-pattern) într-un �ir(text)) 
 
A. Introducere (punerea problemei) 
 
Vom numi sub�irul  a c�rui apari�ie se caut� model iar �irul în care se caut� text.  
Fie  

T[1..n] – textul de lungime n 
P[1..m] – modelul de lungime m, m≤n 

Textul �i modelul sunt, de obicei �iruri de caractere.  
 
Defini�ie. Spunem c� P apare cu deplasamentul s în T dac�: 

1. 0 ≤ s ≤ n-m 
2. T[s+j]=P[j], ∀j, 1 ≤ j ≤ m 

 
Problem�. S� se determine toate deplasamentele s cu care P apare în T. 
 
Exemplu. 
Fie  

T ≡ abcabaabcabac 
P ≡ abaa 

Pentru s=3 se ob�ine 
abcabaabcabac 
   abaa 

 
B. Algoritmul elementer (naiv) 
 
Presupune parcurgerea tuturor deplas�rilor posibile �i verificarea potrivirii modelului în text. 
 
Algoritm: 
 n   lungime(T) 
 m   lungime(P) 
 pentru s   0, n-m execut� 
  dac� P[1..m] = T[s+1..s+m] atunci 
   scrie “am g�sit sub�irul în pozi�ia ”, s+1 
Timp de execu�ie: 
 O((n-m+1)*m) deoarece 
• exist� n-m+1 valori posibile pentru s 
• pentru fiecare s testarea P[1..m] = T[s+1..s+m]implic� un ciclu cu m pa�i 
 
Program simplu: 
 
Program Elementar; 
 
Var T, P: String; 
    n, m, s, j: Byte; 
    f, g: Text; 
    OK: Boolean; 
 

s=3 

Begin 
  Assign(f, 'NAIV.IN'); Reset(f); 
  Assign(g, 'NAIV.OUT'); ReWrite(g); 
  ReadLn(f, T); 
  ReadLn(f, P); 
  n := Length(T); 
  m := Length(P); 
  For s := 0 To n-m Do 
    Begin 
      OK := TRUE;    
      j := 1; 
      While j<=m Do 
        Begin 
          If P[j]<>T[s+j] Then OK := FALSE; 
          Inc(j) 
        End; 
      If OK Then WriteLn(g, 'Am gasit subsirul in pozitia ', s+1) 
    End; 
  Close(g) 
End. 

 
Algoritmul este ineficient deoarece nu folose�te deloc informa�iile ob�inute prin prelucrare 
pentru o valoare oarecare a lui s în prelucr�rile ulterioare. 
 
Exemplu: 
Fie  

T ≡ aaabaab... 
P ≡ aaab 

Se observ� c� pentru s=0 se ob�ine o c�utare cu rezultat pozitiv, dar, în acela�i timp, din 
studierea lui P se observ� c� nici unul din deplasamentele s=1, 2, 3 NU va furniza un 
rezultat pozitiv deoarece P[4]= ‘b’, deci nu are rost s� fie testate aceste valori pentru 
deplasament. 
 
C. Algoritmul Knuth-Morris-Pratt (KMP)  
 
Ideea algoritmului este aceea de a defini o func�ie (numit� “func�ie prefix”) care s� re�in� 
informa�ii depre potrivirea modelului cu deplasamentele lui.  
Func�ia este definit� printr-un tablou cu m elemente, tablou care re�ine în elementul de indice q 
o valoare care indic� ce deplasament se va testa în continuare �tiind c� primele q elemente din 
model “se potrivesc” cu q elemente din text pentru deplasamentul curent. 
 
Formalizând: 
�tiind c� 
 T[s+1..s+q] = P[1..q], deci q elemente se “potrivesc” 
care este cel mai mic deplasament s’>s astfel încât 
 T[s’+1..s’+k] = P[1..k], cu s+q = s’+k 
În cel mai bun caz vom avea s’=s+q, ceea ce indic� faptul c� deplasamentele 
 s+1, s+2, ..., s+q-1 
nu trebuie testate. 
 
Tabloul care define�te func�ia se ob�ine prin compararea modelului P cu el însu�i, astfel: 

• deoarece T[s’+1..s’+k] face parte din por�iunea de text cunoscut�, el este sufix 
al �irului P[1..q] 

• urm�rim exemplul: 

T ≡ bacbababaabcbab 
P ≡ ababaca 
 
T    bacbababaabcbab  s=4 
P              ababaca   q=5 
 
 
T    bacbababaabcbab  s’=6 
P                  ababaca   k=3 
  
 
P[1..5]  ababa   P[1..3] este sufix a lui P[1..5]  
P[1..3]    aba 
 
adic� cel mai lung prefix a lui P care este �i sufix a lui P[1..5] este P[1..3], deci în 
tabloul care reprezint� func�ia vom avea  
 Urm[5]=3 
cu interpretarea: 
“dac� la deplasamentul s s-au potrivit cu succes 5 caractere, urm�torul 
deplasament posibil corect este  

s’ = s + (5 – Urm[5]) = s + (5 - 3) = s + 2” 
• Deci: 

Urm : {1, 2, ..., m} → {0, 1, ..., m-1} 
Urm[q] = max{k, k<q �i P[1..k], care este prefix a lui P, este �i sufix a lui 
P[1..q]} = “lungimea celui mai lung prefix a lui P care este sufix a lui 
P[1..q]” 

 
Algoritm: 
 n   lungime(T) 
 m   lungime(P) 
 Urmatorul(P) 
 q   0 
 pentru i   1, n execut� 
  cât_timp (q>0) �i (P[q+1]<>T[i]) execut� 
   q    Urm[q] 
  dac� P[q+1] = T[i] atunci 
   q  q + 1 //am mai gasit un caracter corect 
  dac� q=m atunci 

scrie “am g�sit sub�irul în pozi�ia ”, i-m+1 
   q    Urm[q] 
 

procedura Urm�torul(P) 
 m   lungime(P) 
 Urm[1]   0 
 k   0 
 pentru q   2, m execut� 
  cât_timp (k>0) �i (P[k+1]<>P[q]) execut� 
   k    Urm[k] 
  dac� P[k+1] = P[q] atunci 
   k  k + 1 
  Urm[q]   k 

s=4 

q=5 

s’=6 

k=3 

Exemplu: 
Cosider�m modelul 

P ≡ ababababca 
S� construim pas cu pas vectorul Urm 
q P   P[1..q]  k Obs 

1 ababababca a   0 (!k<q) 
2 ababababca ab   0 
3 ababababca aba   1 
4 ababababca abab   2 
5 ababababca ababa  3 
6 ababababca ababab  4 
7 ababababca abababa  5 
8 ababababca abababab  6 (!k<q) 
9 ababababca ababababc  0 
10 ababababca ababababca 1 
 
Timp de execu�ie: 
 O(n+m) deoarece 
• calculul tabloului Urm (func�ia prefix) consum� O(m) 
• determinarea “potrivirilor” consum� O(n) 
 
Program simplu: 
 
Program KMP; 
 
Const nMax = 255; 
      mMax = 255; 
 
Var Urm: Array[1..mMax] Of Byte; 
    T, P: String; 
    n, m, q, i: Byte; 
    f, g: Text; 
 
Procedure Urmatorul(P: String); 
Var k, m, q: Byte; 
Begin 
  m := Length(P); 
  Urm[1] := 0; 
  k := 0; 
  For q := 2 To m Do   
     {determina lungimea celui mai lung prefix al lui P 
      care este si sufix al subsirului P[1..q] } 
    Begin 
      While (k>0) And (P[k+1]<>P[q]) Do k := Urm[k]; 
      If P[k+1]=P[q] Then Inc(k); 
      Urm[q] := k 
    End 
End; 
 
Begin 
  Assign(f, 'KMP.IN'); Reset(f); 
  Assign(g, 'KMP.OUT'); ReWrite(g); 
  ReadLn(f, T); 
  ReadLn(f, P); 
  n := Length(T); 
  m := Length(P); 

  Urmatorul(P); 
  q := 0; 
  For i := 1 To n Do 
    Begin 
      While (q>0) And (P[q+1]<>T[i]) Do q := Urm[q]; 
      If P[q+1]=T[i] Then Inc(q);   {s-a mai gasit un caracter corect} 
      If q=m Then 
        Begin 
          WriteLn(g, 'Am gasit subsirul in pozitia ', i-m+1); 
          q := Urm[q] 
        End 
    End; 
  Close(g) 
End. 

 
Dou� exemple de rulare pentru acela�i text T, fi�ierul KMPFIS.PAS, care are 68 linii �i 
dimensiune 1453 bytes, �i dou� modele P (‘ Do’ �i ‘Do’): 
 
Model ‘ Do’ 
Am gasit subsirul in linia 18 pozitia 18 
Am gasit subsirul in linia 21 pozitia 37 

Am gasit subsirul in linia 40 pozitia 19 algoritm naiv 
Am gasit subsirul in linia 46 pozitia 20 
Am gasit subsirul in linia 48 pozitia 39 
Am facut 3453 comparatii 
 
Am gasit subsirul in linia 18 pozitia 18 
Am gasit subsirul in linia 21 pozitia 37 

Am gasit subsirul in linia 40 pozitia 19  algoritm KMP 
Am gasit subsirul in linia 46 pozitia 20 
Am gasit subsirul in linia 48 pozitia 39 
Am facut 1708 comparatii 
 
Vectorul Urm este în acest caz: 

Urm: (0,0,0,0,0,0,0,...) 
 
Model ‘Do’ 
Am gasit subsirul in linia 18 pozitia 19 
Am gasit subsirul in linia 21 pozitia 38 
Am gasit subsirul in linia 40 pozitia 20 
Am gasit subsirul in linia 46 pozitia 21 
Am gasit subsirul in linia 48 pozitia 40 
Am facut 2424 comparatii 
 
Am gasit subsirul in linia 18 pozitia 19 
Am gasit subsirul in linia 21 pozitia 38 
Am gasit subsirul in linia 40 pozitia 20 
Am gasit subsirul in linia 46 pozitia 21 
Am gasit subsirul in linia 48 pozitia 40 
Am facut 1274 comparatii 
 

Vectorul Urm este în acest caz: 
Urm: (0,0,0,0,0,0,0,...) 

 
D. Tem� (programe în care se poate utiliza acest algoritm) 

Problema desert, propus� la barajul pentru lotul na�ional, 2001, Bac�u, autor: �tefan Radu, 
student Universitatea "Transilvania" Bra�ov 

De�ert  

Fi�iere surs�: desert.pas sau desert.c sau desert.cpp 
 Fie o imagine alb-negru care reprezint� fotografia, realizat� din satelit, a unei zone dintr-un de�ert 
unde se caut� o construc�ie secret� de form� dreptunghiular�. Imaginea por�iunii de de�ert analizate a fost 
codificat� într-o matrice având dimensiunile maxime de N × 255 elemente. Imaginea construc�iei este 
codificat� într-o matrice având K × 32 elemente. Elementele celor dou� matrice pot fi numai caracterele 
'.' �i '#'. 
 
Cerin�� 
 Determina�i de câte ori se reg�se�te fotografia construc�iei pe hart�.  
 
Date de intrare 
Fi�ier de intrare: DESERT.IN 
Linia 1: N K 

• dou� numere naturale nenule, N reprezentând num�rul de linii al matricei care codific� fotografia 
de�ertului, iar K num�rul de linii al  matricei p�tratice care codific� fotografia construc�iei; 

Urm�toarele K linii con�in în form� codificat� matricea fotografiei construc�iei: (caractere '#' �i '.' 
neseparate prin spa�ii) 
  c11 c12 ... c1,32 
  c21 c22 ... c2,32 
  . . . . . 

  cK1 cK2 ... cK,32 
Urm�toarele N linii con�in în form� codificat� matricea fotografiei de�ertului (caractere '#' �i '.' ne-
separate prin spa�ii): 
   D1,1 D1,2 ... D1,255 
  D2,1 D2,2 ... D2,255 
  . . . 

  DN,1 DN,2 ... DN,255 
 
Date de ie�ire 
Fi�ier de ie�ire: DESERT.OUT 
Linia 1: NR  

• num�r natural, reprezentând num�rul de potriviri al matricei C în matricea D. 
 
Restric�ii �i preciz�ri 

• 3 ≤ N ≤ 1024 
• 1 < K < N 

 
Exemplu 
DESERT.IN     
3 2  
#............(în total 31 de puncte)... 
#............(în total 31 de puncte)... 
#...........(în total 254 de puncte)....................... 

#...........(în total 254 de puncte).......................  

#...........(în total 254 de puncte)....................... 

DESERT.OUT 
2 

 
Timp maxim de executare/test: 1 secund� 
 
Programul surs�: 
{ autor: �tefan Radu, student Universitatea "Transilvania" Bra�ov } 
{$S+,T-,E-,F-,A+,N+,R-,D+,V+,B-,G+,I+,L+,O-,P-,Q-,X+,Y+} 
{$M 65000,0,655360} 
var f: text; 

    q, x, y, z, n, nc: integer; 
    v: array[0..1024] of longint; 
    dyn: array[0..1024] of integer; 
    poz: array[0..223] of integer; 
    count, l1: longint; 
    s: string; 
 
function s2l(s: string): longint; 
var l: longint; 
    i: integer; 
begin                          {codifica constructia pe biti} 
   l := 0; 
   for i := 1 to 32 do         {1 rand din constructie = 32 biti = 1 longint} 
     begin 
       l := l shl 1; 
       if s[i]='#' then l := l or 1; 
     end; 
   s2l := l; 
end; 
 
begin 
   assign(f, 'desert.in'); reset(f); 
   readln(f, n, nc); 
   for q := 1 to nc do         {citesc cele nc linii ale constructiei} 
     begin  
       readln(f, s); 
       if s='' then break; 
       v[q] := s2l(s);     {si le codific ca nc longint-ti in vectorul v}   
     end; 
   for x := 2 to nc do     {se construieste vectorul 'Urm' din teorie in vectorul dyn}  
     begin 
       y := dyn[x-1];             
       if v[x]=v[y+1] then  
         dyn[x] := dyn[x-1]+1  
       else  
         if v[x]=v[1] then dyn[x] := 1; 
     end; 
 
   for q := 1 to n do      {ia pe rand cele n linii ale fotografiei} 
     begin 
       readln(f, s);       {citesc o linie} 
       if s='' then break; {am terminat}  
       l1 := s2l(s);       {codific primele 32 caractere } 
       inc(n);               
       for x := 0 to 223 do{parcurge cele 255-32=223 caractere} 
         begin 
           y := poz[x];    {scoate in y pozitia caracterului, este si contor} 
           if v[y+1]=l1 then {verific daca se potriveste l1 cu paternul din v}  
             begin         {daca DA} 
               inc(y);       {am mai gasit o potrivire - o numar} 
               if y=nc then  {daca a gasit nc potriviri => am detectat o constructie} 
                 begin  
                   inc(count);  {o numar si incep sa caut}  
                   y := dyn[y]; {de la urmatorul "deplasament posibil corect"}  
                 end; 
             end  
           else            {daca NU}  
             {caut urmatorul "deplasament posibil corect"}  
             while (y>0) and (v[y+1]<>l1) do y := dyn[y]; 
           poz[x] := y;         {si il pun in vectorul poz} 
           l1 := l1 shl 1;      {ma deplasez spre dreapta in s si} 
           l1:=l1 or byte((x<223) and (s[x+33]='#')); {codific ultimul bit} 
        end; 
     end; 
   close(f); 
   assign(f, 'desert.out'); rewrite(f); 
   writeln(f, count); 
   close(f); 
end. 


